
	WWARN is committed to supporting efficient and quality data collection and analysis for antimalarial drug research. As such we are sharing the NONMEM code used for this publication. NONMEM is a computer program, NONlinear Mixed Effects Modeling, used by WWARN’s Pharmacometric Scientific Group. 
For further information, please contact info@wwarn.org 

	Please use the following reference:  
[bookmark: _GoBack]WWARN NONMEM code file used to develop: Population Pharmacokinetic Properties of Piperaquine in Falciparum Malaria: an Individual Participant Data-level Meta-analysis. 
R.M. Hoglund, L. Workman, M.D. Edstein, I. Zongo, J.B. Ouedraogo, S. Borrmann, L. Mwai, C. Nsanzabana, R.N. Price, P. Dahal, N.C. Sambol, S. Parikh, F. Nosten, E.A. Ashley, A.P. Phyo, K.M. Lwin, R. McGready, N.P.J. Day, P.J. Guerin, N.J. White, K.I. Barnes, J. Tarning. 
PLoS Medicine. 2017 Jan 10;14(1):e1002212
Read the full summary of the Piperaquine PK / PD Study Group

	NONMEM code: 
$INPUT     
          ID                          ; Patient ID
	TIME	; Time of sample
	DV                         ; Dependent variable (natural logarithm of observed concentration, ng/mL)
	WT	; Body weight (covariate) 
	EVID	; Event ID record
	MDV	; Missing dependent variable (1 = missing)
	AMT	; Dose amount (mg)
	CMT	; Compartment 
	AGE	; Age (covariate, years)
	MAT	; Matrix (covariate; 1 = capillary, 2 = venous)
	OCC	; Dose occasion (1 = first dose, 2 = second dose , 3 = third dose) 

$DATA  
          dataset.csv    IGNORE = #

$SUBROUTINE  
          ADVAN5 TRANS1

$MODEL      
          COMP = (1)	; Dose
	COMP = (2)	; Central compartment 
	COMP = (3)	; First peripheral compartment
	COMP = (4)	; Second peripheral compartment
	COMP = (5)	; Transit compartment 1
	COMP = (6)	; Transit compartment 2
$PK
;------------------------------------Dose occasion covariate----------------------------------------------------------------------
          F1COVD = (1 + THETA(10) * (OCC - 1))                          ; Linear covariate relationship for dose occasion 
;----------------------------------------------------------------------------------------------------------------------------------------
;------------------------------------Age covariate-----------------------------------------------------------------------------------
          EM50 = THETA(11)                                                                                   ; Age to reach 50% of full maturation
	HILL 	= THETA(12)	; Hill coefficient
	MF 	= ((AGE)**HILL) / (((EM50)**HILL) + ((AGE)**HILL) )	; Age-maturation relationship 
;----------------------------------------------------------------------------------------------------------------------------------------
;------------------------------------ Scaling factor for venous to capillary concentration ----------------------------------
          IF(MAT.EQ.2)     SCALEVC = 1                                                              	; Capillary concentration
	IF(MAT.EQ.1)     SCALEVC = 1 + THETA(9)	; Scaling for venous concentration
;----------------------------------------------------------------------------------------------------------------------------------------
;------------------------------------ Inter-dose occasion variability (IOV) ----------------------------------------------------
          	OC1 = 0	; Initial value, first dose occasion
	OC2 = 0	; Initial value, second dose occasion
	OC3 = 0	; Initial value, third dose occasion
	IF(OCC.EQ.1) OC1 = 1	; First dose occasion
	IF(OCC.EQ.2) OC2 = 1	; Second dose occasion
	IF(OCC.EQ.3) OC3 = 1	; Third dose occasion
	IOV1 = ETA(9) * OC1 + ETA(10) * OC2 + ETA(11) * OC3	; IOV for relative bioavailability
	IOV2 = ETA(12) * OC1 + ETA(13) * OC2 + ETA(14) * OC3	; IOV for mean transit time
;----------------------------------------------------------------------------------------------------------------------------------------

          TVCL 	= THETA(1) * MF * ((WT / 54)**0.75)	; Population clearance
	CL 	= TVCL * EXP(ETA(1))	; Individual 

	TVV2 	= THETA(2) * ((WT / 54)**1)	; Population central volume 
	V2 	= TVV2 * EXP(ETA(2))	; Individual central volume

	TVQ1	= THETA(3) * ((WT / 54)**0.75)	; Population inter-compartment clearance 1
          Q1 	= TVQ1 * EXP(ETA(3))	; Individual inter-compartment clearance 1

          TVV3 	= THETA(4) * ((WT / 54)**1)	; Population peripheral volume 1
	V3 	= TVV3 * EXP(ETA(4))	; Individual peripheral volume 1 

          TVQ2 	= THETA(5) * ((WT / 54)**0.75)	; Population inter-compartment clearance 2
          Q2 	= TVQ2* EXP(ETA(5))	; Individual inter-compartment clearance 2

          TVV4 	= THETA(6) * ((WT / 54)**1)	; Population peripheral volume 2
	V4 	= TVV4 * EXP(ETA(6))	; Individual peripheral volume 2

          TVMT 	= THETA(7)	; Population mean transit time
	MT 	= TVMT * EXP(ETA(7) + IOV2)	; Individual mean transit time

          TVF1 	= THETA(8) * F1COVD	; Population relative bioavailability
          F1 	= TVF1 * EXP(ETA(8) + IOV1)	; Individual relative bioavailability

          NN 	= 2	; Number of transit compartments
          KTR 	= (NN + 1) / MT	; Transit rate constant
	K15 	= KTR	; Transit rate constant (COMP 1 --> 5)
          K56 	= KTR	; Transit rate constant (COMP 5 --> 6)
          K62 	= KTR	; Transit rate constant (COMP 6 --> 2)
          K20 	= CL / V2	; Elimination rate constant (COMP 2 --> 0)
          K23 	= Q1 / V2	; Distribution rate constant (COMP 2 --> 3)
          K32 	= Q1 / V3	; Distribution rate constant (COMP 3 --> 2)
          K24 	= Q2 / V2 	; Distribution rate constant (COMP 2 --> 4)
          K42 	= Q2 / V4	; Distribution rate constant (COMP 4 --> 2)
          S2 	= V2 /1000	; Scaling for central volume
$ERROR
          IPRED = A(2) / S2                                                                      ; Predicted plasma concentration
	IPRED = IPRED *  SCALEVC	; Scaling factor (venous to capillary)
          IF(IPRED.GT.0) IPRED = LOG(IPRED)	; Natural logarithm of predictions
          W = SQRT(SIGMA(1,1))	; Residual error
          IRES = IPRED – DV	; Individual residual error 
	IWRES = IRES / W	; Individually weighted residual error 
	Y = IPRED + EPS(1) 	; Additive residual error 
$THETA                                                                                                    ; Initial estimates of theta
           (0, 55.4)	; 1. Clearance
           (0, 2910)	; 2. Central volume of distribution
           (0, 310)	; 3. Inter-compartment clearance 1
           (0, 4910)	; 4. Peripheral volume of distribution 1
           (0, 105)	; 5. Inter-compartment clearance 2
           (0, 30900)	; 6. Peripheral volume of distribution 2
           (0, 2.11)	; 7. Mean transit time
           (1 FIX)	; 8. Relative bioavailability
           (0, 1.06)	; 9. Scaling factor (venous to capillary)
           (-1, 0.237)	; 10. Dose occasion on relative bioavailability
           (0, 0.575)	; 11. Age to reach 50% of full maturation
           (0, 5.51) 	; 12. Hill coefficient
$OMEGA	; Initial estimates for omega
           (0.075) 	; 1. IIV clearance
           (0.371)	; 2. IIV central volume of distribution
           (0 FIX)	; 3. IIV Inter-compartment clearance 1
           (0.056)	; 4. IIV peripheral volume of distribution 1
           (0.054)	; 5. IIV inter-compartment clearance 2
           (0.114)	; 6. IIV peripheral volume of distribution 2
           (0.135)	; 7. IIV mean transit time
           (0.158)	; 8. IIV relative bioavailability
$OMEGA BLOCK(1) 0.252	; 9. IOV relative bioavailability
$OMEGA BLOCK(1) SAME	;10. IOV relative bioavailability
$OMEGA BLOCK(1) SAME	;11. IOV relative bioavailability
$OMEGA BLOCK(1) 0.195	; 12. IOV mean transit time
$OMEGA BLOCK(1) SAME	; 13. IOV mean transit time
$OMEGA BLOCK(1) SAME	; 14. IOV mean transit time

$SIGMA	; Initial estimates of sigma
           (0.115)	; Residual variability 

$ESTIMATION   POSTHOC MAXEVAL = 9999   METHOD = 1 INTER 

